```
> restart:with(linalg):
```

```
Warning, the protected names norm and trace have been redefined and
unprotected
```

EvEV013.mws, 10-01-29. Set up eigenvalue matrix for problem on left hand side of Fig. 6, p.12 of . . . /∼paul/StLe61ft.pdf.

```
> EVM:=matrix(2,2,[Ixx-lambda,-Ixy,-Ixy,Iyy-lambda]);
```

$$EVM := \begin{bmatrix} Ixx - \lambda & -Ixy \\ -Ixy & Iyy - \lambda \end{bmatrix}$$

Take the determinant to get the characteristic equation eveq.

```
> eveq:=collect(det(EVM),lambda);
```

$$eveq := \lambda^2 + (-Ixx - Iyy)\,\lambda + Ixx\,Iyy - Ixy^2$$

Calculate Ixx, Iyy, Ixy for the 4 given points on LHS of Fig. 6.

```
> Ixx:=0^2+24^2+0^2+(-24)^2;Iyy:=(-25)^2+7^2+25^2+(-7)^2;Ixy:=(-25*0)+(
> 7*24)+(25*0)+(-7)*(-24);
```

$$Ixx := 1152$$
$$Iyy := 1348$$
$$Ixy := 336$$

Hence the numerical form of eveq in this case.

```
> eveq;
```

$$\lambda^2 - 2500\,\lambda + 1440000$$

Solve for the two eigenvalues. Notice how to get the subscripted values out separately.

```
> lambda:=solve(eveq);lambda[1];lambda[2];
```

$$\lambda := 1600, 900$$
$$1600$$
$$900$$

Eigenvector associated with lambda[1]=1600, the larger one, is obtained using either one of the two linearly dependent equations that produce the ratio e2:e1, the slope of this eigenvector.

```
> EVcEq1:=(Ixx-lambda[1])*e1+(-Ixy)*e2;EVcEq2:=-Ixy*e1+(Iyy-lambda[1])*
> e2;
```

$$EVcEq1 := -448\,e1 - 336\,e2$$
$$EVcEq2 := -336\,e1 - 252\,e2$$

Both equations are solved to show that they produce identical results. e2:e1=-4:3 or 4:-3, if you prefer. Another way to look at these numbers is to consider them direction numbers of the eigenvector which is the same as coefficients of the equation of a line on G, the centroid of the given points, and normal to this eigenvector, i.e., e1*x+e2*y+0=0. The eigenvector associated with the larger eigenvalue is parallel to the axis on G that sustains the largest moment of inertia when the given points of unit mass are spun about it. This eigenvector is normal to and defines the line, on G, that fits the points best in the sense of sum of least squares of normal distances.

```
> e2f:=solve(EVcEq1,e2);e2s:=solve(EVcEq2,e2);
```

$$e2f := -\frac{4\,e1}{3}$$

$$e2s := -\frac{4\,e1}{3}$$

The smaller eigenvalue generates the eigenvector normal to the one associated with the larger eigenvalue. Spinning the point cloud about an axis on G parallel to the smaller eigenvalue's eigenvector results in the smallest possible moment of inertia. Now try to reproduce the results of the example on the right of Fig.6 and find the two planes that establish the best fit *line* for the six-point three dimensional example on pp.13-15. If you are really ambitious, try the circle fit example in .../~paul/CaC47t.pdf.

```
> EVcEq12:=(Ixx-lambda[2])*e1+(-Ixy)*e2;EVcEq22:=-Ixy*e1+(Iyy-lambda[2]
> )*e2;
```

$$EVcEq12 := 252\,e1 - 336\,e2$$

$$EVcEq22 := -336\,e1 + 448\,e2$$

```
> e2F:=solve(EVcEq12,e2);e2S:=solve(EVcEq22,e2);
```

$$e2F := \frac{3\,e1}{4}$$

$$e2S := \frac{3\,e1}{4}$$